Introduction to Computer Science - Exercise 2

1

```java
/** Calculate the sum of the square roots of the elements of an array.
 * @param a the array
 * @return Sum (i = 0 to a.length-1) Math.sqrt(a[i])
 */

/**
 * @author Ishmael Aqsar
 *
 */
public class CSArray {

   public static double sumSqrt(double[] a) {
      double sum = 0;
      for (int i = 0; i < a.length; i++) {
         sum += Math.sqrt(a[i]);
      }
      return sum;
   }

   public static void main(String[] args) {
      double[] a = { 1, 2, 3 };
      System.out.println(sumSqrt(a));
   }

}
```

2

```
Compiled from "CSArray.java"
public class CSArray {
  public CSArray();
    Code:
       0: aload_0
       1: invokespecial #8                  // Method java/lang/Object."<init>":()V
       4: return

  public static double sumSqrt(double[]);
    Code:
       0: dconst_0
       1: dstore_1
```

```
 2: iconst_0
 3: istore_3
 4: goto          19
 7: dload_1
 8: aload_0
 9: iload_3
10: daload
11: invokestatic #16                 // Method java/lang/Math.sqrt:(D)D
14: dadd
15: dstore_1
16: iinc          3, 1
19: iload_3
20: aload_0
21: arraylength
22: if_icmplt    7
25: dload_1
26: dreturn
}
```

3

aload_0	slot 0	double[] a
dload_1	slot 1	double sum
iload_3	slot 3	int i

4

$$\text{double sum} = 0; \begin{cases} 0 : \text{dconst\_0} \\ 1 : \text{dstore\_1} \end{cases}$$

$$\text{int i} = 0; \begin{cases} 2 : \text{iconst\_0} \\ 3 : \text{istore\_3} \end{cases}$$

$$\text{for loop;} \begin{cases} 4 : \text{goto 19} \end{cases}$$

$$\text{sum += Math.sqrt(a[i]);} \begin{cases} 7 : \text{dload\_1} \\ 8 : \text{aload\_0} \\ 9 : \text{iload\_3} \\ 10 : \text{daload} \\ 11 : \text{invokestatic \#16} \end{cases}$$

$$\text{i++;} \begin{cases} 16 : \text{iinc 3,1} \end{cases}$$

$$\text{i < a.length;} \begin{cases} 19 : \text{iload\_3} \\ 20 : \text{aload\_0} \\ 21 : \text{arraylength} \\ 22 : \text{if\_icmplt 7} \end{cases}$$

$$\text{return sum;} \begin{cases} 25 : \text{dload\_1} \\ 26 : \text{dreturn} \end{cases}$$

5

Bytecode	Stack
0: dconst_0	0.0
1: dstore_1	empty
2: iconst_0	0
3: istore_3	empty
4: goto 19	empty
7: dload_1	sum
8: aload_0	sum, a
9: iload_3	sum, a, i
10: daload	sum, a[i]
11: invokestatic #16	sum, Math.sqrt(a[i])
14: dadd	sum + Math.sqrt(a[i])
15: dstore_1	empty
16: iinc 3, 1	empty
19: iload_3	i
20: aload_0	i, a
21: arraylength	i, a.length
22: if_icmplt 7	empty
25: dload_1	sum
26: dreturn	empty